
chapter 2

Essential F# 

Topics not yet  covered: Local mutables, Forward references to Reference cells, Arrays

Editors Note: there are a couple of minor issues that readers have spotted with the draft chapter vis.a.vis the current F# release (1.1.12.5) 
· When using the F# Interactive console window with #light code you MUST currently place ";;" on a SEPARATE line, and not at the end of the line as shown in the book.  When using it with Visual Studio you don't need the ";;" at all as you can just highlight the text to execute and hit ALT-RETURN. 
· F# Interactive does not print all values in exactly the same way as shown in the book .  For example, variables bound by a "let" binding only have their types printed, and not their values.  You have to type in the name of a value as an expression to actually see its value printed. 
In this and the next chapter we present all the major constructs of F# by example. This chapter focuses on the simple F# constructs that are used by almost all F# programmers. By the end of this chapter you’ll also have seen examples of many important libraries, and you’ll soon be able to create and manipulate data, use .NET libraries, declare new data types and structure your code using modules, types, interfaces, classes and members. 

Our First F# Program

By now you should have downloaded and installed a version of the F# distribution as described in the introduction. Here’s our first complete F# program. You may not follow it all at first glance, but we’ll explain it piece by piece.

Code Sample CODE-WORD-COUNT - Analyze a string for duplicate words 

#light

/// Analyze a string for duplicate words 

let wordCount text =

    let words = String.split [’ ’] text 

    let wordSet = Set.of_list words 

    let nWords = words.Length 

    let nDups  = words.Length - wordSet.Size 

    (nWords,nDups)

let showWordCount text =

    let nWords,nDups = wordCount text 

    printf "--> %d words in the text\n" nWords

    printf "--> %d duplicate words\n" nDups

We can paste this program into F# Interactive (fsi.exe), remembering to enter ;; to terminate the interactive entry:

> <paste in the above program here> ;;

val wordCount : string -> int * int

val showWordCount : string -> unit

Here F# Interactive has reported the “type” of wordCount and showWordCount – more about types in a moment. Also, sometimes F# Interactive will also show a little more information than we show in this book (for example, some internal details of the generated values) – if trying out these code snippets then you can just ignore that. For now let’s just use the wordCount function interactively:

> let (nWords,nDups) = wordCount "All the king’s horses and all the king’s men";;
val nWords : int 
val nDups : int 
> nWords;;
val it : int = 9

> nDups;;
val it  : int = 2


The above shows the results of executing the function wordCount and binding its two results to the names nWords and nDups respectively, showing that the given text contains 9 words and 2 duplicates.  showWordCount prints the results instead of returning them as a value:

> showWordCount "Couldn’t put Humpty together again";;

--> 5 words in the text

--> 0 duplicate words

From the output we can more or less see what the code does. Now we’ve done that, let’s go through the program in detail.

2.1.1. Turning on the lightweight syntax option

The first line of the file simply turns on the F# lightweight syntax option.  This option is assumed throughout this book, and you should have #light at the head of all your source files.

#light

...

This option allows us to write code that looks and feels simpler simply by omitting recurring F# tokens such as in, ; (semicolon), ;; (double semi-colon), begin and end.  The option instructs the F# compiler and F# Interactive to use the indentation of F# code to determine where constructs start and finish. The indentation rules are very intuituive, and are discussed in Appendix A, anlong with the fully qualified F# syntax.  

Double semi-colons (;;) are still required to terminate entries to F# Interactive, though interactive development environments such as Visual Studio will in general add this automatically.  We show them in the interactive code snippets used this book, though not in the larger samples.

It is generally recommended that you use 4-space indentation for F# code.  TAB characters may not be used in #light code and the F# tools will give an error if they are encountered.
2.1.2. Documenting Code using XMLDocs

The first real line of the program is not code, but a comment. 

/// Analyze a string for the existence of duplicate words.

let wordCount text =

Comments are either lines starting with // or blocks enclosed by (* and *). Comment lines beginning with three slashes /// are “XMLDoc” comments and can, if necessary, include extra XML tags and markup. The comments from a program can be collected into a single .xml file and processed with additional tools, or can be converted immediately to HTML by the F# compiler itself.

2.1.3. Using “let”

The next two lines of the program introduce the start of the definition of the function wordCount and define the local value words, both using the keyword let. 

/// Analyze a string for duplicate words.

let wordCount text =

    let words = ...

let is the single most important keyword you’ll use in F# programming: it is used to define data, computed values, functions and procedures. The left of a let binding is often a simple identifier, but may also be a function name followed by a list of argument names, as in the case of wordCount, which takes one argument text.  The right of a let binding (after the =) is an expression. 

WHY “VALUE”? 

In other languages a local “value” is called a local variable. However in F# you can’t change the immediate value of locals after they’ve been initialized, unless the local is explicitly marked as mutable. Thus F# programmers and the language specification tend to prefer the term “value” to “variable”.

Data indirectly referenced by an immutable local value may still be be mutable, e.g. an immutable local value that is a handle to a hash table may not be changed to refer to a different table, but the contents of the table itself may be altered by invoking operations that add and remove elements from the table.
2.1.4. Types and Type inference

F# is a typed language, so it’s reasonable to ask what the type of wordCount is, and indeed F# Interactive has shown it already:

val wordCount : string -> int * int

This indicates that wordCount takes one argument of type string, and returns int * int, which is F#’s way of saying “the function returns a pair of integers”. No explicit type has been given in the program for wordCount or its argument text, as the full type for wordCount has been “inferred” from its definition. Type inference is discussed further in the side panel and in more detail in later chapters.

Types are very significant in both F# and .NET programming more generally, for reasons that range from performance to coding productivity and interoperability.  Types are used to help structure libraries, to guide the programmer through the complexity of an API, and to place constraints on code to ensure it can be implemented efficiently. However, unlike many other typed languages the type system of F# is both simple and powerful, using orthogonal, composable constructs such as tuples and functions to form succinct and descriptive types. Some of the most important type constructors are shown below.  

Some important types, type constructors and their corresponding values

Type/Type Constructor
Description
Examples

int
Type of 32-bit integers
int


_ option
F# option types
int option,  option<int>
_ list
F# list types
int list, list<int>


_ -> _ 
Function types
int -> int
(_ * ... * _) 
Tuple types
int * int * string 
_ []
Array types
int[]
'a, 'b
Variable types
'a, 'b, 'Key, 'Value
Some type constructors such as list and option are “parameterized”, and instantiations of parameterized types can be written using either prefix notation int list or postfix notation list<int>.
HOW DOES TYPE INFERENCE WORK? 

Type inference works by analyzing your code to collect constraints. These are collected over the scope of particular parts of your program, e.g. each file for the F# command line compiler, and each chunk entered into F# Interactive. These constraints must be consistent, thus ensuring your program is well-typed, and you’ll get a type error if not. Constraints are collected top-to-bottom, left-to-right, outside-in, which is important because long identifier lookups, method overloading and some other elements of the language are resolved using the partial normalized constraint information available at a particular program point.

2.1.5. Calling Functions

Functions are at the heart of most F# programming, and it’s not surprising that the first thing we do is call a library function, in this case String.split. 

let wordCount text =

    let words = String.split [’ ’] text 

    ...

The function String.split takes two arguments. The library reference in Chapter CHAP-LIBREF reveals the type of String.split to be:

val split: char list -> string -> string list

Let’s investigate String.split by running F# Interactive:

> String.split [’ ’] "hello world";;

[ "hello"; "world" ]

> String.split [’a’;’e’;’i’;’o’;’u’] "hello world";;

[ "h"; "ll"; " w"; "rld" ]

We can see that String.split breaks the given text into words using the given characters as whitespace. It is common in F# code to write multiple-argument functions in a style where the arguments come sequentially after the function name, separated by spaces. This is mostly a stylistic choice, but also means functions can be “partially applied” to fewer arguments, leaving a residue function, a useful technique we’ll look at more closely in Section SEC-PARTIAL-APP.

In the above code you can also see examples of


*
Literal characters such as ’ ’and ’a’

*
Literal strings such as "hello world"

*
Literal lists of characters such as [’a’;’e’;’i’;’o’;u’]

*
Literal lists of strings such as the returned value [ "hello"; "world" ].

Literals and lists are covered in detail later in this chapter. Lists are a very important data structure in F#, and we’ll see many examples of their use in this book.  

WHAT IS “STRING” IN “STRING.SPLIT”?

The name String references the F# module Microsoft.FSharp.MLLib.String in the library FSharp.MLLib.dll. This contains a set of simple operations assocaited with values of the string type. It is common for types to have a separate module that contains associated operations.  Some other functions to manipulate values of this type can be found under the path System.String.

Modules under Microsoft.FSharp.MLLib can be referenced by simple prefixes such as String. If you have opened the namespace System you will also have access to functionality in the System.String class via this prefix. 

2.1.6. Using Data Structures

The next portion of the code runs as follows:

...

let wordCount text =

    let words = String.split [’ ’] text 

    let wordSet = Set.of_list words 

    ...

This gives our first taste of the use of data structures from F# code, and the second line lies at the heart of the computation performed by wordCount: it uses the function Set.of list from the F# library to convert the given words to a concrete data structure which is, in practice, much like the mathematical notion of a set, though in reality it is a data structure based internally on trees. We can see the results of converting data to a set by using F# Interactive:

> Set.of_list ["b";"a";"b";"b";"c" ];;

val it : string set = set [ "a"; "b"; "c" ]

> Set.to_list (Set.of_list ["abc"; "ABC"]);;

val it : string list = [ "ABC"; "abc" ]

Here we see several things:


* F# Interactive prints the contents of structured values such as lists and sets;


* Duplicate elements are removed by the conversion;


* The elements in the set are ordered;


* The default ordering on strings used by sets is case-sensitive.

2.1.7. Using Properties and the “.” Notation 

The next two lines of the wordCount function compute the result we’re after – the number of duplicate words. This is done by using two properties Length and Size of the values we’ve computed:

  let wordSet = Set.of_list words in

  let nWords = words.Length in

  let nDups = words.Length - wordSet.Size in

  ...

The properties associated with a value can be found in the type definition associated with the type of the value. F# performs some resolution on property names at compile-time (or interactively when using F# Interactive, where there is no distinction between compile-time and run-time). This is done using static knowledge of the type of the expression on the left of the dot — in this case words and wordSet. Sometimes a type annotation is required in your code in order to resolve the potential ambiguity amongst possible property names. For example, the following code uses a type annotation to note that Length refers to the property associated with list types:

let length (l : ’a list) = l.Length

Type annoations can be useful documentation and when needed should generally be added at the point where a variable is declared.

As can be seen by the use of the dot-notation, F# is both a functional language and an object-oriented langauge. In particular, properties are a kind of “member”, a general term used for any functionality associated with a type. Members referenced by prefixing a type name are called “static members”, and members associated with a particular value of a type are called “instance members”, i.e. instance members are accessed through an object on the left of the dot. See page NNN for more details, and the distinction betweeen values, properties and methods is discussed in the side panel on page XXX.

Sometimes explicitly named functions play the role of members. For example, we could have written the above code as:

    ...

    let nWords = List.length words in

    let nDups = List.length words - Set.size wordSet in

    ...

You will see both styles in F# code. Some F# libraries don’t use members at all, or only sparingly. However, the judicious use of members and properties can greatly reduce the need for trivial get/set functions in libraries, make client code much more readable, and lets programmers who use environments like Visual Studio easily and intuitively explore the primary features of libraries you write.

2.1.8. Using Tuples

The final part of the wordCount function returns the results nWords and nDups as a “tuple”.  

  ...

    let nWords = words.Length 

    let nDups  = words.Length - wordSet.Size 

    (nWords,nDups)

We encountered tuples briefly in our first example and in the context of pattern matching, and they are the simplest but perhaps the most useful of all F# data structures. A tuple value is simply a number of values grouped together to form a new value.

let site1 = ("www.cnn.com",10)

let site2 = ("news.bbc.com",5)

let site3 = ("www.msnbc.com",4)

let sites = (site1,site2,site3)

Here the inferred types of site1 and sites are:

val site1 : string * int

val sites : (string * int) * (string * int) * (string * int)

Tuples can be decomposed into their consituent components in two ways. Firstly, for pairs you can explicitly call the functions fst and snd which, as their abbreviated names imply, extract the first and second parts of a pair:

let url = fst site1

let relevance = snd site1

This gives the values and types:

val url : string = "www.cnn.com"

val relevance : int = 10

The functions fst and snd are defined in the F# library and are always available for use by F# programs – here are their simple definitions:

let fst (a,b) = a

let snd (a,b) = b

More commonly tuples are decomposed using “patterns”, as in the following alternative way of writing the same code:

let url, relevance = site1

let site1,site2,site3 = sites

In this case the names in the tuples on the left of the definitions are bound to the respective elements of the tuple value on the right, so again url gets the value "www.cnn.com" and relevance the value 10.

Tuple values are typed, and strictly speaking there are an arbitrary number of families of tuple types, one for pairs holding two values, one for triples holding three values etc. This means that if you try to use a triple where a pair is expected then you’ll get a type-checking error before your code is run:

let a,b = (1,2,3);;

error: this pattern matches values of type ’int * int’ but is here used with values of type ’int * int * int’. The tuples have different lengths.

Tuples are often used to return multiple values from functions, as in the wordCount example above.  They are also often used for multiple arguments to functions, and frequently the tupled output of one function becomes the tupled input of another function. Here is an example, which shows a different way of writing the showWordCount function we defined and used above:

let showResults (nWords,nDups) =

    printf "--> %d words in the text\n" nWords

    printf "--> %d duplicate words\n" nDups

let showWordCount text = showResults (wordCount text)

showResults accepts a pair as input, decomposed into nWords and nDups, matching the results of wordCount. 

2.1.9. Using Imperative Code

The showWordCount and showResults functions defined in the previous section output the results using a library function called printf:

    printf "--> %d words in the text\n" nWords

    printf "--> %d duplicate words\n" nDups

Those familiar with OCaml will recognise this immediately, and to C/C++ programemrs it will look familiar. Here the pattern %d is a placeholder for an integer, and the rest of the text is output verbatim to the console. Unlike C/C++, printf is a type-safe text formatter, where the F# compiler checks that the subsequent arguments match the requirements of the placeholders. There are a number of other interesting features of printf such as %a and %O placeholders which we’ll encounter later in this book. There are also other ways to format text with F#. For example, we could have used the .NET libraries directly:

    System.Console.WriteLine("--> {0} words in the text\n",nWords)

    System.Console.WriteLine("--> {0} duplicate words\n",nDups)

Here {0} acts as the placeholder, though no checks are made that the arguments match the placeholder before the code is run. We’ll return to the .NET libraries in the next section. The uses of printf also shows how we can use sequence expressions to cause effects in the outside world. The primitive language construct being used here is two expressions separated by ;. The first expression is evaluated, normally for its side-effects, its result discarded, and the overall expression evaluates to the result of the second. Here is a simpler example of this construct:

let two = (printf "Hello World\n"; 1+1)

let four = two + two

When executed this code will print Hello World precisely once, when the right-hand-side of the definition of two is executed. The code from the script could in theory be parenthesized with a semicolon added to make the primitive constructs involved more apparent:

  (printf "--> %d words in the text\n" nWords;

   printf "--> %d duplicate words\n" nDups)

Note that ; is used to sequence code within expressions, and ;; is used to terminate interactions with the F# Interactive session.  Semicolons are optional when the #light lightweioght syntax option is used.

Using .NET Libraries from F#

The true value of F# lies not just in what you can do inside the language, but what you can connect to outside the langauge. For example, F# does not itself come with a GUI library. Instead, F# is connected to .NET, and via .NET to most of the significant programming technologies available on major computing platforms. To emphasize this, our second sample uses two of the powerful libraries that come with the .NET Framework: System.Net and System.Windows.Forms.  The full sample is shown below, and is a script for use with F# Interactive.

Sample HTTP-CODE: Using the .NET Framework Windows Forms and Networking libraries from F#

open System.Windows.Forms

let form = new Form()

form.Visible <- true

form.TopMost <- true

form.Text <- "Welcome to F#"

let textB = new RichTextBox()

textB.Dock <- DockStyle.Fill

form.Controls.Add(textB)

textB.Text <- "Here is some initial text"

open System.IO

open System.Net

/// Get the contents of the URL via a web request

let http(url: string) =

    let req = System.Net.WebRequest.Create(url) 

    let resp = req.GetResponse() 

    let stream = resp.GetResponseStream() 

    let reader = new StreamReader(stream) 

    let html = reader.ReadToEnd() 

    resp.Close()

    html

let google = http("http://www.google.com")

textB.Text <- http("http://news.bbc.co.uk")

2.1.10. Using “open” to Access Namespaces and Modules

The first thing we see in the sample is the use of open to access functionality from the namespace System.Windows.Forms.

open System.Windows.Forms


This simply means we can access any content under this namespace without quoting the long path. If we had not used open we would have to write:

let form = new System.Windows.Forms.Form()

form.Visible <- true

...

which is obviously a little verbose.  open can also be used to access the contents of an F# “module” such as Microsoft.FSharp.MLLib.String without using long paths.

MORE ABOUT OPEN 

Using open is an easy way to access the contents of namespaces and modules. However there are some subtleties. Firstly, open doesn’t actually load or reference a library — instead it reveals functionality from already-loaded libraries (loading libraries using #r or command-line switches is discussed in §X). Libraries and namespace are orthogonal concepts: multiple libraries can contribute functionality to the same namespace, and each library may contribute functionality to multiple namespaces. Often one particular library contributes most of the functionality in a particular namespace. For example, most of the functionality in the System.Windows.Forms namespace comes from a library called System.Windows.Forms.dll. As it happens, this library is automatically referenced by F#, which is why we haven’t needed an explicit reference to the library so far.

In an earlier example, we saw that String in String.split referenced a value in the module Microsoft.FSharp.MLLib.String. By default all F# code is interpreted with an implicit open (§X) of the namespaces Microsoft.FSharp and Microsoft.FSharp.MLLib. If two namespaces have types, sub-namespaces and/or modules with identical names then when you open these the contents of both can be accessed using the same shortened paths. For example, the namespace System contains a type String, and the namespace Microsoft.FSharp.MLLib contains a module String. In this case, long identifier lookups such as String.split search the values and members under both of these, preferring the most recently opened if there is an ambiguity.

Finally, if you ever have name collisions, you can define your own short aliases for modules and types, e.g. by module MyString = My.Modules.String, and type SysString = System.String. 

2.1.11. Using “new” and Setting Properties

The next lines of the sample script use the keyword new to create a top-level window (called a “form”) and set it to be visible. If you run this code in F# Interactive you will see a top-level window appear with the title text “Welcome to F#”.

open System.Windows.Forms

let form = new Form()

form.Visible <- true

form.TopMost <- true

form.Text <- "Welcome to F#"

...

new is shorthand for calling a function associated with the type System.Windows.Forms.Form that constructs a value of the given type — these functions are called “constructors”, and they can only be accessed by using new. Not all F# and .NET types use constructors – you will also see values being constructed using names like Create or via one or more functions in a related module such as String.create or Array.init. We’ll see examples of eachthroughout this book.

A form is an “object” in the sense of the terminology on page XXX, i.e. its properties change during the course of execution, and it is a handle that mediates access to external resources (the display device, mouse etc.) Sophisticated objects like forms often need to be configured, either by passing in configuration parameters at construction or by adjusting properties from their default values after construction. The next lines set the Visible, TopMost and Text properties of the form. When you run the code in F# Interactive you will see the title of the form change as you set the Text property. Not all properties of objects are configurable (i.e. “mutable”) – indeed most are not.

The object we created was bound to the name form. Binding this value to a new name doesn’t create a new form, rather two different handles now refer to the same object (they are said to “alias” the same object).  For example, the following code sets the title of the same form, despite the fact it is being accessed via a different name. 

let form2 = form

form2.Text <- "F# Forms are Fun"

...

The next part of the sample creates a new RichTextBox control and stores it in a variable called textB. A control is typically an object with a visible representation, or more generally, an object that reacts to operating system events related to the windowing system. A form is one such control, but there are many others. A RichTextBox control is one that can contain formatted text, much like a word processor.

let textB = new RichTextBox()

form.Controls.Add(textB)

textB.Dock <- DockStyle.Fill

...

VALUES AND OBJECTS 

In F# everything is a value. In other languages everything is an object. In practice, you can use the words largely interchangeably, though F# programmers tend to reserve “object” for special kinds of values:


* values whose observable properties change as the program executes, usually through the explicit mutation of underlying in-memory data or through external state changes;


* values that refer to data or state that reveal an identity, e.g. a unique integer stamp or the underlying .NET object identity, where that identity may differ from otherwise-identical values;


* values that may be queried to reveal additional functionality, through the use of casts, conversions and interfaces.

F# thus supports objects, but not all values are referred to as objects. Identity and mutation are discussed further on page NNN.

OBJECTS, IDENTITY AND MUTATION 
F# encourages the use of objects whose logical identity (if any) is based purely on the characteristics (e.g. fields and properties) of the object. For example, for essentially all purposes the identity of a pair of integers (1,2) is determined by the two integers themselves: two tuple values both containing these two integers are for practical purposes indistinguishable. This means that tuples support a form structural identity.

F# generally encourages the use of mutation-free data structures where possible. Mutable data structures are permitted, and not all mutable values necessarily reveal their identity through that mutation. For example, sometimes mutation is used just to “bootstrap” a value into its initial configuration (e.g. when connecting the nodes of a graph). Often mutation is used for internal, otherwise invisible state, e.g. a cache. In typical F# programming truly mutable objects are used more rarely than in other settings, though they are used for file handles, graphics objects, graphs and data structures such as hash tables and arrays.

Many of the other traditional concerns of object-oriented (OO) programming are orthogonal to the above distinctions. For example, higher-level programming techniques such as interfaces, inheritance and patterns such as publish/subscribe stem from the OO tradition, and techniques such as functions, type abstraction and functorial operations such as “map” and “fold” from the value-oriented tradition. None of these techniques have any fundamental relationship to mutation and identity: for example interfaces and inheritance can be used very effectively in the context of value-oriented programming. Much of the success of F# lies in the way that it brings the techniques of OO programming and value-oriented programming comfortably together.

XML HELP IN VISUAL STUDIO 

In a rich editor such as Visual Studio 2005 we can easily find out more the functionality in .NET libraries by hovering our mouse over the identifiers in our source code. For example, if we hover over Dock in textB.Dock we will see the XML help shown below.

2.1.12. Fetching a Web Page

The second half of the sample uses the System.Net library to define a function http to read HTML web pages. We can investigate the operation of the implementation of the function by the following lines into F# Interactive.

> open System

> open System.IO

> open System.Net

> let req = WebRequest.Create("http://www.microsoft.com");;

val req : WebRequest

> let resp = req.GetResponse();;

val resp : WebResponse

> let stream = resp.GetResponseStream();;

val stream : Stream

> let reader = new StreamReader(stream) ;;

val reader : StreamReader

> let html = reader.ReadToEnd();;

val html : string = "<html>...</html>"

> textB.Text <- html;;

The final line will set the contents of the text box form to the HTML contents of the Microsoft home page. Let’s take a look at this code line-by-line.

The first line of the code creates a WebRequest object using the “static” method Create, a member of the type System.Net.WebRequest. The difference between “static” and “instance” is discussed in the side panel on page PPP. The result of this operation is an object that acts as a handle to a running request to fetch a web page — we could, for example, abandon the request, or check to see if the request has completed. The second line calls the instance method GetResponse. The remaining lines of the sample get a stream of data from the response to the request using resp.GetResponseStream(), make an object to read this stream using new IO.StreamReader(stream) and read the full text from this stream. We will cover .NET I/O in more detail in Chapter CHAP-TECHNIQUES, but for now we can test by experimentation in F# Interactive that these actions do indeed fetch the HTML contents of a web page. The inferred type for http which wraps up this sequence as a function is shown below.

val http : string -> string

Namespaces, Modules and Type Definitions.

Namespaces, modules and types can all act as containers for values and members. The following table shows the different kinds of containers you will meet as you explore F# and .NET libraries:

Namespace A container for namespaces, modules, types and values. Multiple DLLs can contribute to the same namespace. Examples: System, Microsoft.FSharp
Module A container for nested types, modules and values. Only one DLL may define a particular module. Examples: M’sft.FSharp.MLLib.String, M’sft.FSharp.MLLib.List
Concrete Type A container for members, as well as defining a type constructor in the type system. Only one DLL may define a particular concrete type. Examples:  System.String, System.Int32
Type Abbreviation May not act as a container for additional values or types. The abbreviated path may be used to reference the members contained by a concrete type.  Examples:  string (for System.String), int (for System.Int32)

Values, Methods and Properties

The following table shows the differences between values, methods and properties:

Simple Values Functions, parameters and top-level items defined using let or pattern matching. Examples: form, text, wordCount
Methods Function values associated with types. Interfaces, classes, record and union types may all have associated methods. Methods may be overloaded (see Chapter 3), and must be applied immediately to their arguments. Methods on classes and interfaces may also be abstract (see Chapter 3). Examples: System.Net.WebRequest.Create, resp.GetResponseStream
Properties A short-hand for invoking method members that read or write underlying data. Interfaces, classes, record and union types may all have associated properties. Examples:  System.DateTime.Now, form.TopMost.  
Indexer Properties A property can take arguments, in which case it is an “indexer property”. Indexer properties named Item can be accessed using the .[_] syntax.  Examples:  vector.[3], matrix.[3,4]
STATIC AND INSTANCE

The following table shows the differences between static members and instance members:

Static Items qualified by a concrete type or module.  Examples: System.String.Compare, System.DateTime.Now, List.map, String.split
Instance Methods, properties and values qualified by an expression. form.Visible, resp.GetResponseStream(), cell.contents
The ABC of F# Arithmetic

We now turn to the most common types of data manipulated in F# code. We begin with the basics of F# arithmetic. 

2.1.13. Basic Arithmetic Literals

The table below shows the most common basic types used in F# code and their corresponding literal forms. Many additional types defined in the .NET libraries.

Arithmetic types and literals

Type
Description
Sample Literals
.NET name

bool
true/false values
true, false
System.Boolean
byte
8-bit unsigned integers
0uy, 19uy, 0xFFuy
System.Byte
sbyte
8-bit signed integers
0y, 19y, 0xFFy
System.SByte
int16
16-bit signed integers
0s, 19us, 0x0800us
System.Int16
uint16
16-bit unsigned integers
0us, 19us, 0x0800us
System.UInt16
int, int32
32-bit signed integers
0, 19l, 0x0800, 0b0001
System.Int32
uint32
32-bit unsigned integers
0ul, 19ul, 0x0800ul
System.UInt32
int64
64-bit signed integers
0L, 19L, 0x0800L
System.Int64
uint64
64-bit unsigned integers
0UL, 19UL, 0x0800UL
System.UInt64
single, float32
32-bit IEEE floating-point
0.0f, 19.7f, 1.3e4f
System.Single
double, float
64-bit IEEE floating-point
0.0, 19.7, 1.3e4
System.Double
decimal
High-precision decimal values 
0M, 19M, 19.03M

System.Decimal
bigint
Arbitrarily large integers
0I, 19I
M.F.Math.Types.BigInt
bignum
Arbitrary precision rationals
0N, 19N, 19.03N

M.F.Math.Types.BigNum
2.1.14. Arithmetic Operators

A number of operators are by default overloaded to work over multiple types, including user-defined types. Operators may also be redefined to only work over specific types, by replacing the definition that gives the operator the status of an overloaded operator. See Section SEC-DEFINING-OVERLOADS for a description of how to define new instances of overloads for new type definitions. The most commonly used default arithmetic operators are shown below.

Arithmetic operators and examples

Operator Description
Sample Use on int
Sample Use on float

+
Unchecked Addition
1 + 2
1.0 + 2.0

-
Unchecked Subtraction
12 – 5
12.3 - 5.4
*
Unchecked Multiplication
2 * 3
2.4 * 3.9

/
Division
5 / 2
5.0 / 2.0
%
Modulus
5 % 2
5.4 % 2.0
-
Unary Negation
-(5+2)
-(5.4+2.4)
In F# addition, subtraction and multiplication are unchecked, that is if overflow or underflow occurs beyond the representable range then the tuncated results are returned.  Checked versions of these that raise System.Overflow exceptions can be accessed by opening 
the Microsoft.FSharp.Operators.Checked module.  In all cases division by zero raises a System.DivideByZeroException.  

Operator overloading interacts with type inference — if an overloaded operator is not otherwise constrained to work on any particular type then F# will assume it works on 32-bit integers.

2.1.15. Bitwise Operations

The integer types support bitwise manipulations on their underlying representations. The most commonly used default bitwise manipulation operators are shown in the table below.

Bitwise arithmetic operators and examples

Operator
Description
Sample Use
Result

&&&
Bitwise And
0x65 &&& 0x0F
0x05
|||
Bitwise Or
0x65 ||| 0x18
0x7D

ˆˆˆ
Bitwise Exclusive Or
0x65 ˆˆˆ 0x0F
0x6A

~~~
Bitwise Negation
~~~0x65
0xFFFFFF9a

<<<
Left shift
0x01 <<< 3
0x08
>>>
Right shift (arithmetic if signed)
0x65 >>> 3
0x0C

The following sample shows the use of these operators to encode 32-bit integers into 1, 2 or 5 bytes (the bytes being represented by a list of integers), where integers in the range 0 to 127 occupy one byte.

let encode (n: int32) = 

    if   (n >= 0    && n <= 0x7F)   then [ n ] 

    elif (n >= 0x80 && n <= 0x3FFF) then [ (0x80 ||| (n >>> 8)); (n &&& 0xFF) ]

    else  [ 0xC0; (n >>> 24); ((n >>> 16) &&& 0xFF); 

            ((n >>> 8) &&& 0xFF); (n &&& 0xFF) ]

Here is an example of the function in action:

> encode 32;;

val it : int32 list = [32]

> encode 320;;

val it : int32 list = [129; 64]

> encode 32000;;

val it : int32 list = [192; 0; 0; 125; 0]

2.1.16. Arithmetic Conversions

Numeric types are not implicitly converted — all conversions between different numeric types must be made explicitly. Some conversion operators are shown in the table below.  Often there are multiple equivalent ways of achieving the same conversion.

Basic arithmetic conversions and examples

Operator
Type
Sample Use
Sample Synonyms

Float.of_int
int -> float
Float.of_int 0x65
Int32.to_float, float
Int32.of_float
float -> int
Int32.of_float 128.4
Float.to_int, truncate
Byte.of_int
int -> byte
Byte.of_int 0x65
Int32.to_byte
Int32.of_byte
byte -> int
Int32.of_byte 0x65y
Byte.to_int
These conversions are all unchecked, in the sense that they will not raise exceptions.  An alternative to using these is to use the .NET static methods contained in the type System.Convert, e.g. System.Convert.ToDouble( ).  Uses of System.Convert methods may require type annotations to resolve overloading. In addition, these operators are checked and will raise an exception if the source number can’t be represented within the numeric range of the target type.

2.1.17. Arithmetic Comparisons

When used with numeric data the binary comparison operators =, <>, <, <=, >, >=, min and max perform comparisons according to the natural ordering for each particular numeric type. 

These operators can also be used on other concrete data types, e.g. to compare lists of integers. This topic is discussed in Chapter 4.

Note that when used directly on floating point values these operators implement the IEEE semantics for NaN values, as with nearly all other modern high-level languages.  For example, (NaN = NaN) returns false, as do (NaN <= NaN) and (NaN < NaN).  For more details consult the F# informal specification.

Strings and Basic Formatting

The F# type string is an abbreviation for the.NET Framework type System.String and represents a sequence of Unicode UTF-16 characters.  More details on Unicode can be found in the many excellent guides on the Web. In this section we briefly introduce strings and the most useful sets of functions for formatting them.

2.1.18. String Literals and Primitives

String literals are written using the forms shown in the table below.  

String and character literals

Example

Kind
Type


"Humpty Dumpty"
string 
string
"c:\\Program Files"
string 
string
@"c:\Program Files"
verbatim string
string
"xyZy3d2"B
literal byte string
byte[]
'c'
character 
char
As shown in the table a literal form is also available for interpreting sequences of characters as arrays of bytes.

The escape codes that may be used in strings and characters are shown in the table below. 

Escape characters in non-verbatim strings

Escape

Character
ASCII/Unicode Value
Examples

\n
New Line
10 
"\n"
\r
Carriage Return
13
"\r"
\t
TAB
9
"\t"
\b
Backspace
8


\NNN
Trigraph
NNN
"\032" (space)
\uNNNN
Unicode character
NNNN
"\u00a9" (©)
\UNNNNNNNN
Long unicode character
NNNN NNNN
"\U000000a9" (©)
Verbatim strings are particularly useful for file and path names that contain the escape character “\”.  Multi-line string literals may also be used, e.g. 

let s = "All the kings horses

and all the kings men”;;

val s : string = "All the kings horses\nand all the kings men";;

Strings are immutable, i.e. the contents of a string value may not be modified once created. The operator .[] is used to access the elements of a string, and the property .Length retrieves its length:

> let s = "Couldn't put Humpty";;

val s : string = "Couldn't put Humpty"

> s.Length;;

val it : int = 19

> s.[13];;

val it : char = 'H'

2.1.19. Building Strings 

The simplest way to build strings is via concatenation, using the + operator:

> let s = "Couldn't put Humpty" + " " + "together again”;;

val s : string = "Couldn't put Humpty together again"

Strings can also be built using objects of the .NET type System.Text.StringBuilder. These objects are mutable buffers that can be used to accumulate and modify text. An example is shown below.

> let buf = new System.Text.StringBuilder();;

val buf : System.Text.StringBuilder 

> buf.Append("Humpty Dumpty");;

> buf.Append(" sat on the wall");;

> let s = buf.ToString();;

val s : string = "Humpty Dumpty sat on the wall"

2.1.20. Using printf and Friends

At the start of this chapter we saw a use of the printf function, which is one way to build strings from F# values.  This is a powerful, extensible technique for typesafe formatting.  Here is a second example of a related function for building strings called sprintf:

> sprintf "Name: %s, Age: %d" "Anna" 3;;

val it : string = "Name: Anna, Age: 3"

The format strings accepted by printf and sprintf are recognised and parsed by the F# compiler, and their use is statically type-checked to ensure the arguments given for the formatting holes are consistent with the formatting directives. For example, if you use a integer where a string is expected you will see a type error:

> sprintf "Name: %s, Age: %d" 3 10;;

------------------------------^

error: FS0001: This expression has type int but is here used with type string

Several printf-style formatting functions are provided in the Microsoft.FSharp.MLLib.Printf module.  The most important of these are shown below.

Formatting functions in the Printf module

Function
Outputs via type
Outputs via object
Example

printf
out_channel
stdout
printf "Result: %d" res
eprintf
out_channel
stderr
eprintf "Result: %d" res
fprintf
out_channel
any channel
fprintf stderr "Error: %d" res

twprintf
System.IO.TextWriter
any .NET text writer
twprintf Console.Out "%d" res

sprintf
string
generates strings
sprintf stderr "Error: %d" res

bprintf
System.Text.StringBuilder
any .NET buffer
bprintf buffer "Error: %d" res

The table of basic formatting codes for printf-style formatting is shown below. Format strings used with printf may also include the usual range of specifiers for padding and alignment used by languages such as C.  Full details can be found in the F# library documentation for the Printf module.  

Formatting codes for printf-style string and output formatting

Code

Type accepted
Notes

%b
bool 
prints “true” or “false”

%s
string
%d, %x, %X, %o, 

%ld, %lx, %lX, %lo
int/int32
Decimal/hexadecimal/octal formatting

%ud, %ux, %uX, %uo
uint32
Decimal/hexadecimal/octal formatting

%Ud, %Ux, %UX, %Uo
uint64
Decimal/hexadecimal/octal formatting

%ULd, %ULx, %ULX, %ULo
int64
Decimal/hexadecimal/octal formatting

%e, %E, %f, %g
float
Floating point formats

%M
decimal
See .NET documentation

%O
any type
Format using Object.ToString()

%a
formatter then any type
Format using given formatter
%t
formatter
Run the given formatter
Arbitrary values can be formatted using the %O pattern.  This converts the object to a string using the Object.ToString() function supported by all values.  For example:

> System.DateTime.Now.ToString();;

"28/06/20... 17:14:07"

> sprintf "It is now %O" System.DateTime.Now;;

val it : string = "It is now 28/06/20... 17:14:09"

However, Object.ToString() is a blunt and somewhat undirected way of formatting data.  Structural types such as tuples, lists, arrays and matrices are often poorly formatted by this technique. The F# library defines functions any_to_string and output_any that uses .NET reflection to format any F# value as a string based on the structure of the value.  For example:

> any_to_string (1,2);;

val it : string = "(1, 2)"

> any_to_string [1;2;3];;

val it : string = "[1; 2; 3]"

> printf "The result is %a\n” output_any [1;2;3];;

"The result is [1; 2; 3]"

Structural formatting can be extended to work with any user-defined data types, a topic covered on the F# website. Another extremely useful aspect of the printf formatting functions is their extensibility using %a patterns, which accept both a formatting function and a value to format. This is covered in detail in the F# library documentation for the Microsoft.FSharp.MLLib.Printf module, and an example is given above with the use of %a in conjunction with the output_any function.
Getting Started with Pattern Matching

One of the most important tools in F# programming is “pattern matching”, a very general construct that combines decomposition and control. In the previous sections we had a taste for how pattern matching can be used with tuple values.  However, pattern matching can also be used in many other situations.  We’ll see many other examples of pattern matching in this book, but we’ll begin with some simple pattern matching over strings and integers. Pattern matches on explicit values are introduced using the match ... with ... construct:

let urlFilter url agent =

  match (url,agent) with 

  | "http://www.control.org", 99 -> true

  | "http://www.kaos.org"   , _  -> false

  | _, 86 -> true

  | _ -> false

The inferred type of the function is:

val urlFilter: string -> int -> bool

The expression after match is evaluated to a tuple value, and the “rules” of the match are used to choose a result expression.  Each rule of the match is introduced with a |, though the first | is optional. The first rule of the match will succeed if url and agent are "http://www.control.org" and 99 respectively. The last three rules all make use of “wildcard” patterns represented by the underscore character _  – these match all inputs.

The overall conditions under which urlFilter returns true can be read off simply by reading through the pattern match, e.g. agent 99 may access "http://www.control.org", no one may access "http://www.kaos.org", and, excluding the latter agent 86 may access anything.

2.1.21. Matching on Structured Values

Pattern matching can also decompose values. Here is an example where we match nested tuple values:

let highLow a b =

    match (a,b) with

    | ("lo",lo), ("hi", hi) -> (lo,hi)

    |  ("hi",hi), ("lo", lo) -> (lo,hi)

    | _ -> failwith "expected a both a high and low value";;

The match examines two pairs and looks at the strings in the first element of each, returning the associated values.

> highLow ("hi",300) ("lo",100);;

val it : (int * int) = (100,300)

The first rule decomposes the list according to the case where the first parts of the pairs are the strings "lo" and "hi" respectively. It then returns a pair made from the respective second parts of the tuples. The second rule is the mirror of this in case the values appeared in reverse order. 

The final cases of both of the above examples use wildcard patterns to cover remaining cases. This makes the patterns “exhaustive”.  Frequently no wildcard is needed to ensure this, as for many input types F# is able to determine if the given set of rules is sufficient to cover all possibilities for the given shape of data. However, if a match in not exhaustive a warning is givem, e.g.

> let urlFilter3 url agent =

      match url,agent with 

      | "http://www.control.org", 86 -> true

      | "http://www.kaos.org", _ -> false;;

  match url,agent with 

  ^^^^^^^^^^^^^^^^^^^^

warning: Incomplete pattern match. For example, the value "a",1 will never match any rule.

In these cases it may be necessary to add an extra exception-throwing clause to indicate to the F# compiler that the given inputs are not expected, e.g. 

    ...

    match url,agent with 

    | "http://www.control.org", 86 -> true

    | "http://www.kaos.org", _ -> false;;

    | _ -> failwith "unexpected input"

Non-exhaustive matches are automatically augmented by a default case is added where a MatchFailureException is thrown (exceptions are discussed in Section EXN).

F# is also frequently able to determine if pattern matching rules are redundant, i.e.if a rule may never be selected because previous rules subsume all such cases. In this case a warning is given. For example:

> let urlFilter2 url agent =

      match url,agent with 

      | "http://www.control.org", _ -> true

      | "http://www.control.org", 86 -> true

      | _ -> false;;

     | "http://www.control.org", 86 -> true

     ^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^

warning: This rule will never be matched.

2.1.22. Guarding Rules and Combining Patterns

Individual pattern clauses can be guarded by a condition that is executed if the pattern itself succeeds.  Here is a very simple use of this mechanism to records the three clauses fo computing the sign of an integer:

let sign x =

    match x with 

    | _ when x < 0 -> -1

    | _ when x > 0 ->  1

  | _ -> 0

Two patterns may be combined to represent two possible paths for matching:

let getValue a =

    match a with

    | (("lo" | "low") ,v) -> v

    | ("hi",v) | ("high",v) -> v

    | _ -> failwith "expected a both a high and low value";;

Here the pattern ("lo" | "low") matches either string.  The pattern ("hi",v) | ("high",v) plays essentially the same role by matching pairs values where the left of the pair is "hi" or "high" and binding the value v on either side of the pattern. 

Note: Individual patterns may not bind the same variables twice.  For example, a pattern “(x,x)” is not permitted. Each side of an “or” pattern must bind the same set of variables.
2.1.23. Typecase and null Patterns

Later in this chapter we will see that some F# and .NET types support “subtyping”. For example, essentially every F# type can be coerced to the F# type obj (an abbreviation for the .NET type System.Object). This coercion can be reversed through the use of type tests and casts, and, more generally, typecase and null patterns. We will look at many examples of this in later sections, but for completeness we show the basic form of type case and null-testing patterns in F#:

let switchOnType (a:obj) =

    match a with

  | null                     -> printf "null!"

  | :? System.Exception as e -> printf "An exception: %s!" e.Message

  | :? System.Int32 as i     -> printf "An integer: %d!" i

  | :? System.DateTime as d  -> printf "A date/time: %O!" d

  | _                        -> printf "Some other kind of object\n";

null values

F# types do not support null values except when explicitly modeled using techniques such as option types. However, reference types defined in other .NET languages do support null, and when using .NET APIs you may have to explicitly pass null values to the API using the null keyword and also test return values for null, according to the .NET Framework documentation.  It is recommended that you test for nullnes using null pattern tests, e.g.

    match System.Environment.GetEnvironmentVariable("COMPUTERNAME") with 

    | null -> printf "the environment variable COMPUTERNAME is not defined\n"

    | res -> printf "the environment variable COMPUTERNAME is set to %s\n" res

It is possible to use “backdoor” techniques in the.NET libraries to generate null values for some F# types.  If necessary APIs can check for this condition by first converting values to the obj type by calling box, and then testing for null.  However in practice this is not required by most F# programs.
Working with Lists and Options

Some of the foundational data structures of F# coding are tuples, lists and options. In this section we discuss these and some related topics by example. 

2.1.24. Using F# Lists

F# lists are a common data structures used in foundational F# programming. We have seen some examples of concrete lists when using the String.split function in Code Sample CODE-WORD-COUNT. The following table shows the primitive constructs for building lists:

List related language constructs

Operator
Description
Examples

[] 
The empty list 
[]
[...] 
A list value 
[1; 2; 3]
(::) 
’Cons’ an element with an existing list 
1 :: [2;3]
(@) 
Concatenate two lists 
[1;2] @ [3]
Some basic list values are shown below:

let oddPrimes = [3;5;7;11]

let morePrimes = [13;17]

let primes = 2 :: (oddPrimes @ morePrimes)

F# Interactive shows the value of primes as follows:

val primes : int list = [2;3;5;7;11;13;17]

It is important to note that lists are immutable: the “cons” :: and “append” @ operations do not modify the original lists, instead they create new lists. This can be seen in the following interactive session:

> let people = [ "Adam"; "Dominic"; "James" ];;

val people : string list = [ "Adam"; "Dominic"; "James" ];;

> "Chris" :: people;;

val it : string list = [ "Chris"; "Adam"; "Dominic"; "James" ];;

> people;;

val people : string list = [ "Adam"; "Dominic"; "James" ];;

Note that people has not been changed by the second line – lists and tuples are unchangeable, immutable values.  Immutable values bring many advantages. At first it might seem strange to define values that you can’t change. However, knowing that theses don’t change means that you rarely need to think about the “object identity” of these values – you can pass them to routines and know they won’t be mutated, and you can pass them between multiple threads without worrying about unsafe concurrent access to the values.

As with tuples, you can decompose lists from the head downwards using pattern matching. Here is an example:

let printFirst primes =

    match primes with

    | h :: t -> printf "The first prime in the list is %d\n" h

    | [] -> printf "No primes found in the list\n"

> printFirst oddPrimes;;

The first prime in the list is 3

This code examines the head of the primes list, considering the case where primes is an empty list. Note that the :: and [] symbols can be used both to build up lists in expressions and to decompose them in pattern matching. This applies to many data structures in F# programming, and we’ll discuss this further in § ??.

The F# library also includes a module List that contains some useful values related to programming with lists. We’ll be seeing many of these functions in the next section and throughout this book.  Some of these are shown in Fig. LIST-MODULE.

F# lists are represented in memory as linked lists, and each F# list value is either a ’cons’ cell containing a value plus a pointer to the next chain in the list. When you create a new list using :: the tail of the new list will point to the old list, which ensures that the inner memory associated with lists is often reused as a part of multiple list values. 
F# lists are not appropriate for all circumstances: e.g. very large data structures should probably be represented using arrays or other data structures, or even managed by an external tool such as a relational database.  A number of list-like data structures are discussed in the panel on page PPP.

Figure LIST-MODULE: Some Sample Functions in the List Module

module List

val hd : ’a list -> ’a

val tl : ’a list -> ’a list

val map : (’a -> ’b ) -> ’a list -> ’b list

val filter : (’a -> bool) -> ’a list -> ’a list

val iter : (’a -> unit) -> ’a list -> unit

val to_array : ’a list -> ’a []

val of_array : ’a [] -> ’a list
LISTS AND SOME LIST-LIKE DATA STRUCTURES 

F# lists are simple, but are just one of a number of important list-like data structures that are available with F#. The following table contrasts some of these:


*
int list, Immutable linked lists. Cheap to add and access data from one end. Inefficient for random access lookup.  


*
int[], int array. Mutable, fixed size arrays, intrinsic to .NET, Constant-time access


*
int[,]. Mutable, fixed size two-dimensional arrays, intrinsic to .NET, Constant-time access


*
System.Collections.Generic.IEnumerable<_>.  Abstract view of enumerable objects and collections. See also the functions in the F# module IEnumerable.


*
System.Collections.Generic.List<_>.  Mutable, auto-resizing arrays. Constant-time access


*
System.Collections.Generic.LinkedList<_>.  Mutable, doubly-linked lists. Linear-time access


*
System.Collections.ArrayList.  Mutable, auto-resizing ‘object’ arrays. From an early version of .NET, not often used with F#. 

The Microsoft.FSharp type of immutable lists and the System.Collections.Generic type of expanding arrays both use the name “list”, though the types are very different and their concrete type definitions live in different namespaces. We recommend you use the type abbreviation 

type ResizeArray<’a> = System.Collections.Generic.List<’a> 

to make clear which type is being referred to. For this reason the F# distribution defines a module ResizeArray that provides F#-idiomatic functions to create and transform objects of this type.

2.1.25. Using F# Option Values

Like lists and tuples, option values are simple constructs frequently used as a workhorse in F# coding. An option is simply either a value Some(v), or the absence of a value None. Here is a data structure that uses options to represent the (optional) parents of some well-known characters:

> let people = [ ("Adam", None);

                 ("Eve" , None);

                 ("Cain", Some("Adam","Eve"));

                 ("Abel", Some("Adam","Eve")) ];;

val people : (string * (string *string) option) list

Pattern matching is very frequently used with option values:

> let showParents (nm,parents) = 

      match parents with  

    | Some(dad,mum) -> printf "%s has father is %s, mother %s\n” nm dad mum

    | None          -> printf "%s has no parents!\n";;

val showParents : (string * (string * string) option) -> unit

> showParents ("Adam",None);;

Adam has no parents

The F# library also includes a module Option that contains some useful values related to programming with options. While it is easy to code these by hand using pattern matching it can also be useful to learn and rely on the standard definitions.

Using Function Values

In this section we take a look at the next foundational building block of F# programming: function values.  

2.1.26. Transforming Lists with “List.map”

One of the primary uses of F# lists is as a general-purpose concrete data structure for storing ordered sets of inputs and ordered results. Input sets are often transformed to output sets using “aggregate” operations that transform, select, filter and categorize elements of the list according to a range of criteria.  These aggregate operations provide an excellent introduction to the use of function values. Let’s take a closer look at this in the following code sample, which continues on from the definition of http in code sample HTTP-CODE.

> let sites = [ "http://www.live.com";


                "http://www.google.com" ]

val sites : string list = [ ... ]

> let fetch url = (url, http url);;

val fetch : string -> string * string

>let resultsOfFetch = List.map fetch sites

val resultsOfFetch: (string * string) list

= [ ("http://www.live.com", "<html>...</html>");

    ("http://www.google.com", "<html>...</html>"); ]

 (The actual HTML strings and numbers shown will be different to those shown here.) The first definition is a literal list of URLs. The second line calls the aggregate operation List.map.  This accepts the “function value” fetch as the first argument and a list as the second argument, applying fetch to each element of the list and collecting up the results in a new list. 
Types are one useful way to help learn what a function does.  Here’s the type of List.map:

> List.map;;

val it: ('a -> 'b) -> 'a list -> 'b list

This says that List.map accepts a function value as the first argument and a list as the second argument, and returns a list as the result. The function argument can have any type 'a -> 'b, and the list must have a corredponding type 'b. The symbols 'a and 'b are called “type parameters” or “type variables”, and functions that involve type variables are called “generic” or “polymorphic”. Type parameters are discussed in detail in the next section. 

2.1.27. Anonymous Function Values

Function values are so common in F# programming that it is very convenient to define them without giving them names.  Here is a very simple example:

> let primes = [2;3;5;7];;

val primes : int list = [2;3;5;7]

> let primeCubes = List.map (fun n -> n * n * n) primes;;

val primeCubes : int list = [8;27;125;343]

The definition of notPrime uses the “anonymous function” (fun n -> n * n * n). These are like function definitions but are unnamed and appear as an expression rather than as a let declaration. fun is a keyword meaning “function”, n represents the argument to the function, and n*n*n is the result of the function. The overall type of the anonymous function expression is int -> int.  We could have this technique to avoid defining the intermediary function fetch in the sample above.

let resultsOfFetch = List.map (fun url -> (url, http url)) sites

We will see anonymous functiosn throughout this book.  Here is a final example:

> let sizes = List.map (fun (_,p) -> String.length p) resultsOfFetch;;

val sizes : int list = [3932; 2827 ] 

Here we see two things:


*
The argument of the anonymous function is a tuple pattern. The use of a tuple pattern automatically extracts the second element from each tuple and gives it the name p within the body of the anonymous function.


*
Part of the tuple pattern is a “wildcard” pattern, indicated by an underscore. This indicates that we don’t care what the first part of the tuple is, given we are only interested in extracting the length from the second part of the pair.

2.1.28. More  Aggregate Operators

Aggregate operators such as List.map are powerful constructs, especially when combined with anonymous functions, tuples, functions-as-values and options. Here is a longer example using List.filter and List.map to count the number of URL-links in each of two well-known web pages, and then collect stats on a group of pages:

let lettersAndDigits = [ '0' .. '9' ] @ [ 'a' .. 'z' ] @ [ 'A' .. 'Z' ]

let getWords s = String.split lettersAndDigits s

let stats site =

    let url = "http://" + site 

    let html = http url 

    let hwords = html |> getWords 

    let hrefs = html |> getWords |> List.filter (fun s -> s = "href") 

    (site,html.Length, hwords.Length, hrefs.Length)

let sites = [ "www.live.com";"www.google.com";"search.yahoo.com" ];;

sites |> List.map stats;;

val it : (string * int * int * int) list

  = [("www.live.com", 7728, 1156, 10);

     ("www.google.com", 2685, 496, 14);

     ("search.yahoo.com", 10715, 1771, 38)]

The stats function generates the length of the html for the given website, the number of words in the text of that HTML, and the approximate number of links on that page. 

The above code sample makes extensive use of the |> operator to “pipeline” operations. The F# library design ensures that a common, consistent set of aggregate operations is defined for each structured type. Figure FIG-AGG shows how the same design patterns occur for many different abstractions.

Figure FIG-AGG: A Recurring Aggregate Operator Design Pattern from the F# Library

Operator
Type

List.map
(’a -> ’b) -> ’a list -> ’b list
Array.map
(’a -> ’b) -> ’a[] -> ’b[]
Option.map
(’a -> ’b) -> ’a option -> ’b option
IEnumerable.map
(’a -> ’b) -> #IEnumerable<’a> -> #IEnumerable<’b>
PIPELINING WITH |> AND >>

The |> operator is perhaps the most important operator you use in your F# programming. Its definition is deceptively simple:

let (|>) x f = f x

Here is a use of the operator to compute the cubes of three numbers:

[1;2;3] |> List.map (fun x -> x * x * x)

This produces [1;8;27], just as if we’d written:

List.map (fun x -> x * x * x) [1;2;3]

In a sense |> is just “function application in reverse”. However, the use of |> often has distinct advantages:


*
Clarity. When used in conjunction with well-designed libraries, the |> operator allows us to perform the data transformations and iterations in a forward-chaining, pipelined style.


*
Type inference. The use of the |> operator allows type information to be flowed from input objects through to the functions manipulating those objects. As discussed on page XXX, F# uses information collected from type inference to resolve some language constructs such as property accesses and method overloading. This relies on information being propagated left-to-right through the text of a program. In particular, typing information to the right of a position is not taken into account when resolving property access and overloads.

For completeness, here is the type of the operator:

val (|>) : ’a -> (’a -> ’b) -> ’b

2.1.29. Composing Functions

Above we saw how the operator |> can be used to “pipe” values through a number of functions. This was a small example of the process of “computing with functions”, and essential and powerful programming technique in F#. In this section we look at ways to compute new function values from existing ones using compositional techniques. First we look at function composition, which is one of the simplest ways of generating new values from old. For example, consider the following code:

google |> getWords |> List.filter (fun s -> s = "href") |> List.length;;

We can rewrite this code using function composition as follows:

let countLinks = getWords >> List.filter (fun s -> s = "href") >> List.length;;

google |> countLinks;;

Let’s take a look at this more closely. We have defined countLinks as the composition of three function values using the operator >>. This operator is defined in the F# library as follows:

let (>>) f g x = g(f(x))

We can see from the definition that f >> g gives a function value that first applies f to the x and then applies g. Here is the type of >>:

val (>>) : ('a -> 'b) -> ('b -> 'c) -> ('a -> 'c)

Note that >> is typically applied to only two arguments — those on either side of the binary operator, here named f and g. The final argument x is typically supplied at a later point. F# is very good at optimizing basic constructions of pipelines and composition sequences from functions — for example, the function countLinks above will become a single function calling the three functions in the pipeline in sequence. This means sequences of compositions can be used with relatively low overhead.

2.1.30. Building Functions with Partial Application

Composing functions is just one way to compute interesting new functions. Another useful way is through the use of “partial application”. Here’s an example:

let shift (dx,dy) (px,py) = (px + dx, py + dy)

let shiftRight = shift (1,0)

let shiftUp    = shift (0,1)

let shiftLeft  = shift (-1,0)

let shiftDown  = shift (0,-1)

The last four functions have been defined by calling shift with only one argument, in each case leaving a residue function that expects an additional argument. F# Interactive will report the types as:

val shift : (int * int) -> (int * int) -> (int * int)

val shiftRight : (int * int) -> (int * int)

val shiftLeft  : (int * int) -> (int * int)

val shiftUp    : (int * int) -> (int * int)

val shiftDown  : (int * int) -> (int * int)

Here is an example of the use shiftRight, and also the use of an inline application of shift to new arguments (2,2):

> shiftRight (10,10);; 

val it : (int * int) = (11,10)

> List.map (shift (2,2)) [ (0,0); (1,0); (1,1); (0,1) ];; 

val it : (int * int) list = [ (2,2); (3,2); (3,3); (2,3) ]

2.1.31. Using Local Functions 

Partial application is one way in which functions can be “computed”, rather than simply being defined. This technique becomes very powerful when combined with additional local definitions. Here’s a simple and practical example, representing an idiom common in graphics programming:

open System.Drawing;;

let remap (r1: Rectangle) (r2: Rectangle) =

    let scalex = float r2.Width / float r1.Width 

    let scaley = float r2.Height / float r1.Height 

    let mapx x = r2.Left + truncate (float (x - r1.Left) * scalex) 

    let mapy y = r2.Top + truncate (float (y - r1.Top) * scaley) 

    let mapp (p: Point) = new Point(mapx p.X, mapy p.Y) 

    mapp

The function remap computes a new function value mapp that maps points in one rectangle to points in another. F# Interactive will report the type as:

val remap : Rectangle -> Rectangle -> (Point -> Point)

The type Rectangle is defined in the .NET library System.Drawing.dll and represents rectangles specified by integer coordinates. The computations on the interior of the transformation are perfomed in floating point to improve precision.  We can use this as follows:

> let mapp = remap (new Rectangle(100,100,100,100)) (new Rectangle(50,50,200,200));;

val mapp : Point -> Point

> mapp (new Point(100,100));;

val it : Point = X=50,Y=50

> mapp (new Point(150,150));;

val it : Point = X=150,Y=150

> mapp (new Point(200,200));;

val it : Point = X=250,Y=250

The intermediary values scalex and scaley are only computed once. This is despite the fact that we’ve called the resulting function mapp three times. It may be helpful to think of mapp as a function object being generated by the remap function.

In the above sample: mapx, mapy and mapp are “local functions”, i.e. functions defined locally as part of the implementation of remap. Local functions can be context-dependent, i.e. they can be defined in terms of any values and parameters that happen to be in scope. Here mapx is defined in terms of scalex, scaley, r1 and r2. Local functions are, if necessary, implemented by taking the “closure” of the variables they depend upon and storing them away until needed. In optimized F# code the F# compiler often avoids this implementation and intead passes extra arguments.
Recursion, Loops and Control

So far in this chapter we’ve seen some of the simple but powerful F# data types and language constructs that make up F# programming. In this section we look more closely at a number of constructs for controlling the execution of an F# program.

2.1.32. Defining Recursive Functions

One of the fundamental building blocks of computation in F# is recursion. The following code shows a simple well-known recursive function:

> let rec factorial n = if n <= 1 then 1 else n * factorial (n-1);;

val factorial : int -> int

> factorial 5;;

val it : int = 120

This example shows that a recursive function is simply one that can call itself as part of its own definition. Recursive functions are introduced by let rec. Functions are not recursive by default, since it is wise to minimize the size of a recursive collection of functions to aid with maintainability. It may help to visualize the execution of factorial 5 in the following way (though note that in reality F# executes the function using efficient native code): 

factorial 5

= 5 * factorial 4

= 5 * (4 * factorial 3)

= 5 * (4 * (3 * factorial 2))

= 5 * (4 * (3 * (2 * factorial 1 )))

= 5 * (4 * (3 * (2 * 1)))

= 5 * (4 * (3 * 2))

= 5 * (4 * 6)

= 5 * 24

= 120

As with all calls, the execution of the currently executing instance of the function is effectively suspended while the recursive call is made. An exception is made for “tail-calls”, where the execution of the current instance is effectively completed and a “goto” performed instead, discusses further in Chapter TECHNIQUES.

Many of the operators we have so far encountered can be coded as recursive functions.  For example, the following is one possible implementation of List.map:

let rec map f l =

    match l with

    | [] -> []

    | h :: t -> f x :: map f t

Likewise, many functions such as List.length are implemented using recursive functions.

2.1.33. Conditionals, && and ||

A basic control construct in F# programming is if/then/elif/else.  Here’s an example:

let round x =

    if x >= 100 then 100

    elif x < 0 then 0  

  else x

Conditionals are really shorthand for pattern matching, e.g. the above could have been written:

let round x =

  match () with 

    | _ when x >= 100 -> 100

    | _ when x < 0    -> 0 

    | _               -> x

Conditionals are always guarded by a boolean-valued expression.  These can be built using the && (“and”) and || (“or”) as well any library functions that return boolean values.  

let round2 (x,y) =

    if x >= 100 || y >= 100 then 100,100

    elif x < 0 || y < 0 then 0,0

  else x,y

&& and || are special values that only evaluate the second argument if needed.  

Note: If you don’t use the #light lightweight syntax option, then when you combine conditionals with imperative code you will sometimes need to use parentheses (…) or begin/end to delimit the regions covered by each branch of the conditional, e.g. if x > 100 then begin … end.  If you use the #light syntax then these are optional, as long as the body of the then and else branches are correctly indented from the if and elif tokens of the construct.
2.1.34. Iterating with Loops

Two simple looping constructs are available to help make writing iterative code with side effects dead-simple. The first is a for loop, illustrated here by a replacement implementation of the repeatFetch function above:

let repeatFetch url n =

    for i = 1 to n do

        let html = http url 

        printf "fetched <<< %s >>>\n" html

    done

    printf "Done!\n"

This loop is executed for successive values of i over the given range, including both start and end indexes. The second looping construct is a while loop, which repeats until a given guard is false. For example, here is a way to keep your computer busy until the weekend:

open System

let loopUntilMonday () =

    while (DateTime.Now.DayOfWeek <> DaysOfWeek.Saturday) do

        printf "Still working!\n"

    done

    printf "Saturday at last!\n"

Both the for and while constructs are for writing imperative programs, indicated partly by the fact that both are expressions whose type is unit. Note also that the done token is optional when using the #light lightweight syntax option, though it is considered good style to leave it in.

2.1.35. Iterating with Aggregate Operators

It is very common to use data to drive control, and indeed in functional programming the distinction between data and control is often blurred: function values can be used as data, and data can influence control flow. One example is using a function such as List.iter to iterate over a list.  Let’s take a simple example:

let sites = [ "http://www.live.com";

              "http://www.google.com";

              "http://search.yahoo.com" ]

List.iter (fun site -> printf "%s, length = %d\n" (http site).Length)

List.iter simply calls the given function (here an anonymous function) for each element in the input list. Here is its type:

val List.iter: ('a -> unit) -> 'a list -> unit

Many additional aggregate iteration techniques are available, particularly by using values of type IEnumerable<_>. We cover this in Chapter TECHNIQUES.

Exceptions

When a routine encounters a problem it may respond in several ways, e.g. by recovering internally, emitting a warning, returning a marker value or incomplete result, or by throwing an “exception”. The following code indicates how an exception can be thrown by some of the code we have already been using:

let req = System.Net.WebRequest.Create("not a URL");;

System.UriFormatException: Invalid URI: The format of the URI could not be determined.

Similarly the GetResponse method also used in the http function may raise a System.Net.WebException. The exceptions that may be raised by routines are typically recorded in the documentation for those routines. 

2.1.36. Raising and Catching Execptions

Exceptions can be thrown by using the raise function from the F# library and caught using the try ... with ... language construct, e.g.:

try 

  raise (new System.InvalidOperationException("invalid operation"))

with 

  | :? System.InvalidOperationException -> printf "caught!\n";;

caught!

The following code sample shows how to use try ... with ... to catch two kinds of exceptions arising that may arise from the operations that make up the http method, in both cases returning the empty string "" as the incomplete result.  Note that try ... with ...  is just an expression, and may return a result in both branches:

let http(url: string) =

    try

        let req = System.Net.WebRequest.Create(url) 

        let resp = req.GetResponse() 

        let stream = resp.GetResponseStream() 

        let reader = new StreamReader(stream) 

        let html = reader.ReadToEnd() 

        html

    with 

        | :? System.UriFormatException -> ""

        | :? System.Net.WebException -> ""

Here the with clause of a try ... with ... construct uses a “type test pattern match” to filter the exceptions caught. 

2.1.37. Using try/finally

Exceptions may also be caught using the try ... finally ... construct.  This guarantees to run the finally clause both when an exception is thrown and when the expression evaluates normally.  This should be used when resources must be disposed after the completion of an operation. For example, the inner portion of the above function should be written:

      ...

      try 

          let stream = resp.GetResponseStream() 

          let reader = new StreamReader(stream) 

          let html = reader.ReadToEnd() 

          html

      finally 

          resp.Close();

Tip: Whenever you create objects that have Close or Dispose operations, or which implement the IDisposable interface, you should be sure to consider using try/finally or its related construct Idioms.using to ensure the resources are disposed when the object is no longer accessible.  
2.1.38. Extracting Data from Exceptions

When an exception is thrown an value is created that records information about the exception.  This value may be used in the with clause of try/with constructs.  For example, all exception values support the Message property:

try 

  raise (new System.InvalidOperationException("invalid operation"))

with 

   err -> printf "huh? msg = '%s'\n" err.Message;;

huh? msg = 'invalid operation'

Exception values are always refinements of the F# type exn, an abbreviation for the .NET type System.Exception.  F# also supports an extended kind of exception object that carries data in a conveniently accessible form.  For example, here is a declaration of a new class of exceptions and a function that wraps http with a filter that catches particular cases:

exception BlockedURLException of string

let http2 url = 

    if url = "http://www.kaos.org" 

    then raise(BlockedURLException(url))

    else http url

The information from F# exception values can be extracted, again using pattern matching:

try 

    http2 "http://www.kaos.org" then raise(BlockedURLException(url))

   (new System.InvalidOperationException("invalid operation"))

with 

    | BlockedURLException(url) -> printf "blocked! url = '%s'\n" url;;

blocked! url= 'http://www.kaos.org'

A summary of exception-related language and library constructs is shown in table TAB-EXCEPTION.
Table TAB-EXCEPTION: Exception-related language and library constructs

Example Code
Kind
Notes

raise ...
F# library function
Raises the given exception

failwith ...
F# library function
Raises the FailureException exception

try ... with ...
F# expression
Catches expressions matching the clauses

try ... finally ...
F# expression
Executes the “finally” expression on both when the computation is successful and when an exception is raised

| :? ArgumentException ->
F# pattern rule
A rule matching the given .NET exception type

| Failure(msg) -> 
F# pattern rule
A rule matching the given data-carrying F# exception

| exn -> 
F# pattern rule
A rule matching any exception, binding the name exn to the exception object value. 

| exn when ... -> 
F# pattern rule
A rule matching the exception under the given condition, binding the name exn to the exception object value.

Understanding Generics and Subtyping

F# constructs such as lists, tuples and function values are all “generic”, which means they can be instantiated at multiple different types.  For example, int list, string list and (int * int) list are all instantiations of the generic family of F# list types.  Likewise, int -> int and string -> int are both instantiations of the generic family of F# function types represented by ->.  There are also many other generic types and operations in the F# library and the.NET Framework.  

In this section we look briefly at how F# determines instantiations for generic parameters, and how F# automatically infers generic types for function and member definitions.

2.1.39. Type Variables 

Generic constructs are always represented through the use of “type variables”, which in F# syntax are written 'a, 'b, 'k, 'key, 'K  etc.  For example, the definition of the list type in the F# library begins:

    type 'a list = ...

Values can also be generic. A typical generic value is List.map, whose type is as follows:

    val map : ('a -> 'b) -> 'a list -> 'b list

Each time you name a generic type or value the F# type system must infer instantiations for the type variables involved.  For example, in the previous section we used hen we used “List.map fetch” over an input list. The type of the latter was:

val fetch : string -> string * string

In this case, in the signature of List.map the type variable 'a is instantiated to string and 'b to string * string.  

Sometimes the binding point of these variables will be made explicit in error messages, and explicit bindings can also be used in type annotations, e.g. you may see:

    val map<'a,'b> : ('a -> 'b) -> 'a list -> 'b list

2.1.40. Automatic Generalization

A key feature of F# is the “automatic generalization” of code.  The combination of automatic generalization and type inference makes many programs simple, succinct and general. It also greatly enhances code reuse. Languages without automatic generalization force programmers to compute and explicitly write down the most general type of their functions, and often this is sufficiently tedious that programmers do not take the time to abstract common patterns of data manipulation and control.

For example, type parameters are automatically introduced when writing simple functions that are independent of some of their arguments:

let getFirst (a,b,c) = a

Here the inferred type of getFirst is reported as: 

val getFirst: 'a * 'b * 'c -> 'a

That is, getFirst has been inferred to be generic in three type variables, where the result type is related to the first element of the input tuple. Automatic generalization is applied when a let or member definition doesn’t fully constrain the types of inputs or outputs. You can tell automatic generalization has been applied by the presence of type variables in an inferred type, and ultimately by the fact you can reuse a construct at multiple types.

Automatic generalization is particularly useful when taking functions as inputs.  For example, the following takes two functions as input and applies them to each side of a tuple:

let mapPair f g (x,y) = (f x, g y)

The generalized, inferred type is:

val mapPair : ('a -> 'b) -> ('c -> 'd) -> ('a * 'c) -> ('b * 'd)

the Value Restriction

F# sometimes requires a little help before a definition is automatically generalized.Typically only function definitions are automatically generalized – for historical reasons this is called the “value restriction”.  For example, the following definition does not result in a generic type and gives an error:
let empties = Array.create 100 []

----^^^^^^^

error: FS0030: Value restriction. Type inference has inferred the signature

        val empties : '_a list []

but its definition is not a simple data constant. Either define 'empties' as a simple data term, make it a function, or add a type constraint to instantiate the type parameters.

The code attempts to create an array of empty lists. Here the error message indicates that automatic generalization would give empties the type 'a list []. However, this type makes no sense: it would imply that we’ve created an array of lists somehow suitable for use with any type 'a, when in reality our array should only be usable with one type, e.g. int list [] or string list [], but not both.  (If it were usable with both then we could store an integer list into the array and fetch it out as a string list!)  The “value restriction” ensures that declarations don’t result in this kind of confusion: automatic generalization is not applied to declarations unless they are functions or simple, immutable data constructs.

You can work around this restriction in two ways.  First, if you intended to create a function that generated arrays of different types on demand than you can often make the declaration into a function by adding a dummy argument:

let duplicate x = List.map (fun x -> (x,x)) x

let makeEmpties () = Array.create 100 []

let intEmpties    : int    list [] = makeEmpties()

let stringEmpties : string list [] = makeEmpties()

However, if you really meant to create one value then simply use an explicit type annotation, e.g.

let empties : int list [] = Array.create 100 []

2.1.41. Subtyping

For the most part, programming in F# involves using types in a “simple” way, where types are related through the use of explicit code to map from one type to another.  For example, values of the type int are distinct from values of the type float, and values of one record type such as Person are distinct from values of another such as Company. This is very often sufficient, partly because type inference and function values make it easy to write generic code. However, it is convenient for a language to also support implicit relationships between types.  F# supports these implicit relationships thorugh subtyping. Indeed, subtyping in F# is exactly the same as that used by the .NET Framework, so if you are familiar with another .NET language you’ll already know how things things work.

Let’s explore how sub-typing works by using F# Interactive. Firstly, let’s look at how some of the F# types we’ve already seen relate to the type obj (an abbreviation for the .NET type System.Object):
> let xobj = (1 :> obj);;

val xobj : obj = 1

> let sobj = ("abc" :> obj);;

val sobj : obj = "abc"

Here we are simply investigating the subtyping relationship through the use of the built-in coercion (or “up cast”) operator :>. This operator is used to convert a value to any of its super-types. For some types this operation may involve a change of representation called “boxing”. Boxing may also be performed with the F# library function box.

As with most object-oriented languages, subtyping is reversible. We can see this through the following examples:

> let sobj = box "abc";;

val sobj : obj

> let s = (sobj :?> string);;

val s : string = "abc"

Here we are reversing the coercions performed by :> by using the “down cast” operator :?>. Objects such as xobj and sobj are implicitly annotated with the “runtime type” of the object. The operator :?> will raise an exception if the object is not of a suitable type:

> let xobj = box 1;;

val xobj : obj = 1

> let x = (xobj :?> string);;

error: InvalidCastException raised at or near stdin:(2,0)

The :?> operator is query on the runtime type of an object. A more convenient way of performing these queries is by using pattern matching:
> match (box "abc") with

  | :? string as s -> printf "the input is the string '%s'\n" s

  | :? int as d    -> printf "the input is the integer '%d'\n" d

  | _ -> printf "the input is something else\n"

the input is the string 'abc'

So far we have only investigated subtyping between ordinary types and the type obj. Subtyping occurs between other types as well, in particular:


*
All types are subtypes of System.Object

*
Record and discriminated union types are subtypes of the interface types they implement


*
Interfaces types are subtypes of the other interfaces they extend


*
Class types are subtypes of both the interfaces they implement and the classes they extend


*
Array types are subtypes of the .NET type System.Array

*
Value types (e.g. types such as int32 that are abbreviations of .NET value types) are subtypes of the .NET type System.ValueType. Likewise .NET enumeration types are subtypes of System.Enum.

2.1.42. Combining Subtyping and Generics

F# functions can be declared to accept parameters that are any subtype of a particular type by using type annotations of the form #type, where type is some type that supports subtyping. This is particularly common when working with .NET libraries that use subtyping heavily, or when working with abstractions such as interface values.  For example,

> open System.Windows.Forms

> let setTextOfControl (c : #Control) (s:string) = c.Text <- s;;

val setTextOfControl: #Control -> string -> unit

This notation is shorthand for a generic function and a “constraint” on the type variable, and we could equally have written:

> open System.Windows.Forms

> let setTextOfControl (c : 'a) (s:string) : _ when 'a :> Control = c.Text <- s;;

val setTextOfControl: #Control -> string -> unit

Diving into Type Definitions

The final topic we cover in this chapter is some simple ways to define your own types through type abbreviations, records and discriminated unions. In the next chapter we’ll take a closer look at object-oriented programming in more detail, including defining classes and interfaces.  We will also discuss how to choose and combine these mechanisms effectively in mixed object-oriented/functional programming.

2.1.43. Using Type Abbreviations

Type abbreviations are the very simplest type definitions:

type index = int

type flags = int64

type results = string * TimeSpan * int * int

It is common practice to use lower-case names for type abbreviations. Type abbreviations may be generic:

type 'a stringDictionary = System.Collections.Generic.Dictionary<string,'a>

type ('a,'b) maps = ('a -> 'b) * ('b -> 'a)

Type abbreviations are not “concrete”, as they simply alias an existing type. Type abbreviations are expanded during the process of compiling F# code to the format shared between multiple .NET languages. The difference can, for example, be detected by other .NET languages, and because of this a number of restrictions apply to type abbreviations. For example, you cannot augment type abbreviations with additional members, as can be done for concrete types such as records, discriminated unions and classes (see Chapter 3).  In addition, you cannot truly hide a type abbreviation using a signature (again see Chapter 3).

2.1.44. Defining and Using Records

The simplest concrete type definitions are records. Here’s our first example:

type Person =

    { Name: string;

      DateOfBirth: System.DateTime; }

Record values may be constructed simply by using the record labels:

> let bill = { Name = "Bill"; DateOfBirth = new System.DateTime(1962,09,02) }

val bill : Person = { Name="Bill"; DateOfBirth = 02/09/1962 }

Records values may also be constructed by using the following more explicit syntax, which names the type should there be a conflict between labels amongst multiple records:
> let anna = { new Person

               with Name = "Anna"

               and  DateOfBirth = new System.DateTime(1968,07,23) }

val anna : Person = { Name="Anna"; DateOfBirth = 23/07/1968 }

Record values are often used to return results from functions:
type PageStats =

    { Site: string;

      Time: System.TimeSpan;

      Length: int;

      NumWords: int;

      NumHRefs: int }

This technique works well when returning a large number of heterogeneous results. 

let stats site =

    let url = "http://" + site 

    let html,t = time (fun () -> http url) 

    let hwords = html |> getWords 

    let hrefs = html |> getWords |> List.filter (fun s -> s = "href") 

    { Site=site; Time=t; Length=html.Length; NumHRefs=hrefs.Length }  

Here is the type of stats and how F# Interactive shows the results of applying the function:
val stats : string -> PageStats

> stats "www.live.com";;

{ Site="www.live.com"; Time=0.872623628; Length=7728, NumWords=1156; NumHRefs=10; }

Records labels need not be unique amongst multiple record types. Here is an example:

type Person =

    { Name: string;

      DateOfBirth: System.DateTime; }

type Company =

    { Name: string;

      Address: string; } 

When record names are non-unique, constructions of record values may need to use object expressions in order to indicate the name of the record type, thus disambiguating the construction. For example, consider the following type definitions:

type Dot = { X: int; Y: int }

type Point = { X: float; Y: float }

On lookup, record labels are accessed using the “.” notation in the same way as properties. One slight difference is that in the absence of further qualifying informataion; the type of the object being accessed is inferred from the record label. This is based on that latest set of record labels in scope from record definitions and uses of open. For example, given the above definitions we have:

> let coords1 (p:Point) = (p.X,p.Y)

val coords1 : Point -> int * int

> let coords2 (d:Dot) = (d.X,d.Y)

val coords2 : Dot -> float * float

> let dist p = sqrt (p.X * p.X + p.Y * p.Y)  // use of X and Y implies type “Point”

val dist : Point -> float

The accesses to the labels X and Y in the first two definitions have been resolved using the type information provided by the type annotations. The accesses in the third definition have been resolved using the default intepretation of record field labels in the absence of any other qualifying information.

2.1.45. Using Mutable Record Fields

Record fields may be labelled mutable.  This is usually done for records that implement the internal state of objects, a topic we’ll discuss in the next chapter.  This means that record fields can be updated using the <- operator, i.e. the same syntax used to set a property.

type ConnectionStats =

    { mutable TotalLength: int;

      mutable NumPages: int; }

let fetch cstats url =

    let page = http url 

    cstats.NumPages <- cstats. NumPages + 1

    cstats.TotalLength <- cstats.TotalLength + page.Length

    page

The function fetch takes two arguments: one a mutable “tracker” record used to accumulate statistics and the other the URL to access. Programming with mutable data structures is covered in more detail in Chapter TECHNIQUES.

> let cstats = { TotalLength = 0; NumPages = 0; }

val it : ConnectionStats = { TotalLength = 0; NumPages = 0 }

> ( fetch cstats "http://www.smh.com.au"; 

    fetch cstats "http://www.theage.com.au ";

    cstats);;

val it : ConnectionStats = { TotalLength = 8372; NumPages = 2 }

Finally, record types may also support members (e.g. properties and methods) and give implicit implementations of interfaces, as discussed further below.  This makes them a very powerful device for implementing object-oriented abstractions.
2.1.46. Using Discriminated Unions

The second kind of concrete type definition is a discriminated union. Here is a very simple example:

type Route = int

type Make = string

type Model = string

type Transport =

    | Car of Make * Model

    | Bicycle

    | Bus of Route

Each alternative of a discriminated union is called a “constructor”. Values can be built simply by using the constructor much as if it were a function:

> let nick = Car "BMW 360";;

val nick : Transport

> let don = [ Bicycle; Bus 8 ];;

val don  : Transport list

> let james = [ Car "Ford Fiesta"; Bicycle ];;

val james  : Transport list

Constructors can also be used in pattern matching:

let averageSpeed (tr: Transport) =

    match tr with

    | Car _ -> 35

    | Bicycle -> 16

    | Bus _ -> 24

Several of the types we’ve already met are defined as discriminated unions. For example, the 'a option type is defined as follows:

type 'a option =

    | None

    | Some of 'a

Discriminated unions may include recursive references (the same is true of records and other type definitions). This is frequently used when representing structured languages via discriminated unions:

type Proposition =

    | True

    | And of Proposition * Proposition

    | Or of Proposition * Proposition

    | Not of Proposition

Recursive functions are required to define the semantics of this kind of type. For example:

let rec eval (p: Proposition) =

    match p with 

    | True -> true

    | And(p1,p2) -> eval p1 && eval p2

    | Or (p1,p2) -> eval p1 or eval p2

    | Not(p1) -> not (eval p1)

Indeed, the F# type of immutable lists is defined in this way:

type ’a list =

    | ([])

    | (::) of''a * 'a list

A broad range of tree-like data structures are very conveniently represented as discriminated unions. For example:

type Tree<'a> =

    | Tree of 'a * Tree<'a> * Tree<'a>

    | Tip

Recursive functions can be used to compute properties of trees:

val size: Tree<'a> -> int

let rec size tree = 

    match tree with 

    | Tree(_,l,r) -> 1 + size l + size r

  | Tip -> 1

Here is an example of a constructed tree term and the use of the size function:

> let small = Tree("1”,Tree("2”,Tip,Tip),Tip);;

val small : Tree<string> = Tree ("1",Tree ("2",Tip,Tip),Tip)

> size small;;

val it : int = 2

Symbolic manipulation based on trees is discussed in detail in Chapter SYMBOLIC. Multiple types may be involved in a mutually recursive collection of types, including record types, discriminated unions and abbreviations:

type node = 

    { Name : string; 

      Links : link list }

and link =

    | Dangling

    | Link of node 

Like records, discriminated union may support members and give implicit implementations of interfaces, as discussed further below.
Note: Discriminated unions are a powerful and important construct, and are crucial when modeling a finite, sealed set of choices, as is often the case in applications and symbolic analysis tasks.  However, they are, by design, non-extensible, except where that extensibility is represented by other mechanisms such as object and function values.  Subsequent modules may not add new discriminants to a particular discriminated union. This is deliberate: types such as options and lists are powerful partly because they put a limit on which possibilities exist. Extensibility must be defined through alternative techniques, including the use of records of functions, interfaces and classes. 

�See the notes above for using #light with F# interactive in the current release.


�Note: in a future release F# Interactive will be changed to print out values as soon as variables are bound.


�NYI in F# 1.1.11


�NYI in F# 1.1.11


�This is NYI in F# 1.1.12.  Also a local open is really needed.


�Computing and printing a sample incomplete match is NYI in F# 1.1.11


�F# 1.1.11 does not define this module.  It will be added to the next release of F#.


�Maybe move to “Techniques”


�This syntactic form is NY in F# 1.1.11


�F# 1.1.11 incorrectly shows “null” for some nullary constructors as in this example





